

software

C-128

FONT 80

di Felice Sobrero - Milano

Caratteri programmabili in video a 80 colonne sul C-128

Operando in 80 colonne con il processore video 8563, il set di caratteri non viene letto in ROM, come di norma, ma risiede nella RAM INTERNA del Video Chip dove viene «iniettato» rapidamente ogni volta che accendiamo il C-128: è la routine CHARRAM (\$CE0C Dec. 52748) del sistema operativo che si occupa di questa funzione.

Non è facile mettere le mani su questa zona di memoria separata oltre i 128K, all'interno del VDC, ma noi prepareremo una routine in linguaggio macchina del tutto simile all'originale, che prenderà il set di caratteri da una zona del BANCO 1 (RAM), anziché del BANCO 14 (ROM), per «iniettarla» nel VDC. La chiameremo CHARRAM per analogia.

Per preparare i nostri set di caratteri inizialmente copieremo i dati della ROM nel banco 1, dove potremo editarli con comodo e registrarli su disco insieme alla CHARRAM che servirà per lanciarli. I vantaggi di questo metodo sono principalmente i seguenti:

- 1) durante l'uso di nuovi caratteri non si occupa un solo byte della RAM utente, né nel banco 1 né nel banco 0;
- 2) a meno delle particolari tecniche descritte in aprile da Lonardi e Tagliabue, non è possibile invadere per errore la memoria caratteri;
- 3) non c'è da spostare NESSUN PUNTATORE, il che garantisce la massima compatibilità del nuovo set di caratteri con tutto il software.

Di tutto questo si occupa il programma FONT 80.

Per la routine CHARRAM ho scelto la zona \$1300 (Dec. 4864), poco sotto

l'area del Basic e normalmente inutilizzata: la routine è comunque rilocabile.

Al termine delle operazioni FONT 80 ci permetterà di salvare su disco il set personalizzato in forma di file binario allocato da \$D000 (Dec 53248) in avanti, nonché la routine CHARRAM, sempre in forma di file binario da \$1300.

Notate che l'editing nel banco 1 (buffer) avviene agli indirizzi originari della ROM: \$D000-\$DFFF (Dec. 53248-57343), il che consente, probabilmente, di poter caricare vecchi set di caratteri già definiti per l'uso in 40 colonne anche con il C-64. Inoltre in questo modo se trovate troppo lento e primitivo il mio editor, potrete usare qualche altro programma più potente e con esso registrare il file binario principale.

Una volta creato il nostro set con l'abbinata routine CHARRAM su disco, per installarlo saranno sufficienti queste istruzioni (da tastiera o da programma)

```
BLOAD «nome del set», B1
BLOAD «CHARRAM SYS 4864», B0
SYS 4864
```

e vedremo comparire sullo schermo i nostri caratteri.

A questo punto sia il contenuto del banco 1 sia la routine a \$1300 possono essere cancellati o sovrascritti, perché non servono più!

Commento alle linee del programma

Ho cercato di evitare per quanto possibile i caratteri di controllo per tutti coloro che lo scriveranno da tastiera!

A prezzo di un allungamento delle routine, gli input non rispondono al tasto di RETURN da solo, salvo durante l'edizione del carattere, dove

possiamo confermare semplicemente ciò che deve restare immutato.

1100 controllo del modo 80 colonne.

1330 questo è il cuore di FONT 80: è il byte che indica da quale banco prendere i dati carattere. La routine originale contiene il valore 14.

1380 apertura di uno pseudo-file

Inviare i vostri programmi

Alcuni lettori ci chiedono, nelle loro lettere, come sottoporre i loro programmi a MC.

Registrate i vostri lavori su cassetta o disco (se il programma è proprio molto corto può bastare il semplice listato; certo, la cassetta non guasta mai...), corredateli dell'opportuna documentazione e spedite il tutto alla redazione, indicando magari sulla busta la rubrica interessata.

Tutti i programmi che arrivano sono esaminati ed i migliori pubblicati. Purtroppo non possiamo restituire, per ragioni organizzative, il materiale che ci viene inviato, anche in caso di mancata pubblicazione.

Ricordatevi che migliore è la documentazione, maggiore è la possibilità che il vostro lavoro venga pubblicato: spiegate quindi chiaramente il funzionamento del programma ed accludete tutto quello che pensate possa essere utile (elenco variabili e via dicendo). Soprattutto non dimenticate di indicare il computer sul quale il programma gira, né il vostro nome e indirizzo e, se possibile, il numero di telefono. Indicate anche, per la retribuzione se il programma sarà pubblicato, luogo e data di nascita, domicilio fiscale e codice fiscale (partita IVA, se la possedete).

Il compenso per i programmi pubblicati varia normalmente fra le 40 e le 150.000 lire, a seconda della qualità del lavoro inviato; eventuali programmi di particolare complessità ed interesse potranno essere valutati al di fuori di questo standard, previ accordi con la redazione.

dalla tastiera: serve ad evitare il fastidioso punto interrogativo in input.

1520 memorizzazione nel buffer RAM (1) del carattere ridefinito.

1680 premendo N/X lo schermo adotterà il set standard o quello modificato, per provarlo. Invariato il contenuto del buffer.

1850 si potrebbe vantaggiosamente sostituire con una routine in Assembler, più veloce.

2000 la freccetta indicatrice del menu: ci ricorda se stiamo vedendo il set Commodore o i nuovi caratteri.

2060 (e 2140) routine per convertire in valori numerici 0-255 i caratteri ingranditi che vi mostra l'editor.

2200 scelta di uno dei 4 subset da 1K ciascuno.

2260 qui come altrove POKE 209,0 svuota il buffer di tastiera per evitare «scelte» accidentali.

Nota

I codici di controllo nei listati sono riportati in forma «esplicita», in conseguenza dell'impiego della stampante Star NL-10 e relativa interfaccia per Commodore. Ovviamente, nella digitazione del programma è necessario usare i consueti tasti che corrispondono alle indicazioni fra parentesi: ad esempio cursore destro per (RGHT), CTRL-3 per (RED) eccetera.

(CLR)	=	␣
(HOME)	=	␣
(DOWN)	=	␣
(UP)	=	␣
(RGHT)	=	␣
(LEFT)	=	␣
(BLK)	=	␣
(WHT)	=	␣
(RED)	=	␣
(CYN)	=	␣
(PUR)	=	␣
(GRN)	=	␣
(BLU)	=	␣
(YEL)	=	␣
(RVS)	=	␣
(OFF)	=	␣
(ORNG)	=	␣
(BRN)	=	␣
(LRED)	=	␣
(GRY1)	=	␣
(GRY2)	=	␣
(LGRN)	=	␣
(LBLU)	=	␣
(GRY3)	=	␣
(SWLC)	=	␣

```

1000 REM*****
1010 REM
1020 REM          - FONT 80 -
1030 REM
1040 REM FELICE SOBRERO 1986 - PER MC MICROCOMPUTER
1050 REM
1060 REM*****
1070 REM
1080 REM "(UP)" = CURSORE ↑           "(LEFT)" = CURSORE←
1090 REM
1100 IF RGR(0)<>5 THEN PRINT"PASSARE A 80 COLONNE!":END
1110 SCNCLR:CHAR,72,24,"FONT 80":WINDOW 6,2,79,23,1
1120 H=17:J=16:FAST:BANK 15:SYS 52748:REM VDC ← ROM
1130 SH$="":REM 37
1140 PRINT"PREMERE: 1 - EDITA SET ORIGINALE COMMODORE"
1150 PRINT"                2 - EDITA SET REGISTRATO SU DISCO"
1160 PRINT:PRINT
1170 POKE 209,0:GETA$
1180 IF A$="1"THEN GOSUB 1850:GOTO 1280
1190 IF A$="2"THEN GOSUB 1940:GOTO 1220
1200 GOTO 1170
1210 REM-----
1220 REM CARICAMENTO SET DA DISCO
1230 REM-----
1240 TRAP 1250:DCLEAR:BLOAD(F$),B1,P53248
1250 IF DS>18 THEN BEGIN
1260 PRINTCHR$(7):PRINT DS$:SLEEP1:SCNCLR:GOTO 1140:BEND
1270 REM-----
1280 REM COPIA CHARROM DA $CEOC A $1300
1290 REM-----
1300 BANK15:RESTORE:PRINTCHR$(7)::
1310 FORA=4864 TO 4927
1320 POKEA,PEEK(A+47884):NEXT
1330 POKE 4888,1:REM MODIFICA N.BANK !!
1340 REM-----
1350 REM          EDITOR CARATTERI
1360 REM-----
1370 TRAP 1390
1380 CLOSE 1:OPEN 1,0:REM FILE TASTIERA X INPUT
1390 SCNCLR:GOSUB 2200:PRINT:A$=""
1400 INPUT"(UP) CARATTERE          ":A$:IF A$="" THEN 1400
1410 C%=ASC(A$):SCNCLR:PRINT SPC(12):IF S$>"2" THEN PRINT CHR$(18
);
1420 PRINT CHR$(C%);CHR$(146);" = ASCII":C%:PRINT
1430 U%=(C% AND 128)/2)OR(C% AND 63):REM VIDEO ← ASCII
1440 Y=IN+U%*8:REM PUNTO INIZIO IN MEMORIA
1450 PRINT SH$
1460 FORX=0TO7:BANK1:A%(X)=PEEK(Y+X):BANK15:NEXT
1470 FORX=0TO7:GOSUB 2060:PRINT B$(X):TAB(9):A%(X):NEXT
1480 PRINT"(HOME)":PRINT:PRINT:FOR X=0 TO 7:GOSUB 2160
1490 PRINTTAB(20):B$(X);"(LEFT) (LEFT) (LEFT) (LEFT) (LEFT) (LEFT) (LEFT)
(LEFT)":REM 8 CURSOR←
1500 INPUT#1,B$(X):GOSUB 2160:PRINT TAB(30):A%(X)
1510 NEXT
1520 FORX=0TO7:BANK1:POKE(Y+X),A%(X):BANK15:NEXT
1530 REM TRASFERISCE RAM ← EDITOR
1540 REM-----
1550 REM          MENU' OPZIONI DELL'EDITOR
1560 REM-----
1570 PRINT:PRINT SH$
1580 PRINT"PREMERE: 1 - ALTRO CARATTERE"
1590 PRINT"                2 - RECUPERA SET STANDARD"
1600 PRINT"                3 - USCITA DALL'EDITOR"
1610 PRINT"                N - VEDERE NUOVO SET"
1620 PRINT"                X - VEDERE SET STANDARD"
1630 GOSUB 2000
1640 POKE 209,0:GETA$
1650 IF A$="1" THEN SCNCLR:GOTO 1350

```

```

1660 IF A$="2" THEN SYS 52748:GOSUB 1850:GOTO 1280
1670 IF A$="3" THEN 1720
1680 IF A$="N" THEN GOSUB 2000:SYS 4864:REM VDC ← RAM
1690 IF A$="X" THEN GOSUB 2000:SYS 52748:REM VDC ← ROM
1700 GOTO 1640
1710 REM-----
1720 REM      SALVATAGGI E FINE
1730 REM-----
1740 SCNCLR:DCLEAR:PRINT"SALVATAGGIO SU DISCO ? (S/N)"
1750 GETKEY A$:IF A$>"S" THEN PRINT:PRINT "FINE LAVORO":END
1760 PRINT:PRINT:GOSUB 1940
1770 BSAVE(F$),B1,P53248 TO P57344:REM SALVA SET CARATTERI
1780 IF DS>18 THEN PRINT CHR$(7):PRINT DS$:SLEEP 2:GOTO 1720
1790 R$="CHARRAM SYS 4864"
1800 PRINT:PRINT DS$;" ";F$:SCRATCH(R$)
1810 BSAVE(R$),B0,P4864 TO P4929:REM SALVA ROUTINE
1820 IF DS>18 THEN PRINT CHR$(7):PRINT DS$:END
1830 PRINT:PRINT DS$;" ";R$:END
1840 REM-----
1850 REM SET STANDARD: RAM(1) ← ROM(14)
1860 REM-----
1870 PRINT:PRINT:PRINT:PRINT CHR$(15);
1880 PRINT "ATTENDERE 40'"
1890 FOR A=53248 TO 57344
1900 BANK 14:B%-PEEK(A):BANK 1:POKE A,B%
1910 NEXT:PRINTCHR$(143);
1920 RETURN
1930 REM-----
1940 REM INPUT NOME FILE PER I/O DISCO
1950 REM-----
1960 F$="":PRINT:INPUT"(UP) (UP) NOME DEL FILE ";F$
1970 IF F$="" THEN 1960
1980 F$=LEFT$(F$,16):RETURN
1990 REM-----
2000 REM      INDICATORE STATO
2010 REM-----
2020 IF A$="N" THEN H=16:J=17:ELSE H=17:J=16
2030 CHAR,33,H,"←":CHAR,33,J," "
2040 RETURN
2050 REM-----
2060 REM      CONVERSIONE BYTE A STRINGA
2070 REM-----
2080 W$="":B%-A%(X):FOR K=7 TO 0 STEP-1
2090 IF 2↑K <= B% THEN BEGIN
2100 W$=W$+"*":B%=B%-2↑K:REM BIT PIENO
2110 BEND:ELSE W$=W$+"-":REM BIT VUOTO
2120 NEXT K:B$(X)=W$:RETURN
2130 REM-----
2140 REM      CONVERSIONE STRINGA A BYTE
2150 REM-----
2160 B%=0:W$=B$(X):FOR K=1 TO 8
2170 IF MID$(W$,K,1)="" THEN B%=B%+2↑(8-K)
2180 NEXT K:A%(X)=B%:RETURN
2190 REM-----
2200 REM      SCELTA SET MAIUSCOLO / MINUSCOLO
2210 REM-----
2220 PRINT:PRINT"PREMERE: 1 = SET MAIUSCOLO"
2230 PRINT"          2 = SET MINUSCOLO"
2240 PRINT"          3 = SET MAIUSCOLO REVERSE"
2250 PRINT"          4 = SET MINUSCOLO REVERSE"
2260 POKE 209,0:GETKEY S$
2270 IF S$="1" THEN IN=53248:PRINT CHR$(142):RETURN
2280 IF S$="3" THEN IN=54272:PRINT CHR$(142):RETURN
2290 IF S$="2" THEN IN=55296:PRINT CHR$(14):RETURN
2300 IF S$="4" THEN IN=56320:PRINT CHR$(14):ELSE GOTO 2260
2310 RETURN
2320 REM----- FINE -----


```

MC



**CDS
Informatica
srl**

PACKAGES SU PERSONAL COMPUTER



**Corsi di formazione
per Managers,
Professionisti
e Quadri Aziendali.**

**CORSI PER PRINCIPIANTI
E AVANZATI**

durata 1 o 2 settimane

I corsi sono tenuti dalla CDS Informatica nei locali dell'International Language Centre, in viale Parioli 101c.

Le lezioni vengono tenute adoperando IBM Personal Computers e prevedono un massimo di due studenti per ogni sistema.

Per informazioni telefonare a:

- International Language Centre - t. 802990
- CDS Informatica srl - t. 5405723

oppure ritagliare e spedire il coupon alla CDS Informatica, Piazza Caduti della Montagnola 50 - Roma

Vi prego di mandarmi maggiori dettagli sui vostri corsi di formazione.

Cognome _____
Nome _____
Indirizzo _____
Telefono _____